Mapping properties of exponential function

In this note, we will use Mathematica to visualize some mapping properties of the exponential function $f(z) = e^z$. The methodology explained below is applicable to any complex functions.

Each complex number $z = x + iy$ corresponds to a point at position (x, y) on the plane. Thus, f maps a point to a point. We say that f is a *geometric transformation* on the plane.

Each point A on the vertical line $x = 1$ gets mapped by f to another point (called the image of A under f) on the plane. It is natural to ask: what does the set of all those images as A varies on the line $x = 1$ look like?

The line $x = 1$ has complex parametrization $z = 1 + it$ where $t \in \mathbb{R}$. The plot of the line for $t \in [0, 1]$ is as follows (Figure 1).

```
ParametricPlot[{1, t}, {t, 0, 1}]
```

or equivalently

```
ParametricPlot[ReIm[1 + t*I], {t, 0, 1}]
```

The command **ReIm** returns the list of real part and imaginary part. For example **ReIm[1+I*3]** returns $\{1, 3\}$. To declare function f, we use the command:



Figure 1

```
f[z_] := Exp[z]
```

To plot the image of the line under f, we write (Figure 2)

```
ParametricPlot[ReIm[f[1 + t*I]], {t, 0, 1}, AspectRatio -> Automatic,
 AxesOrigin -> {0, 0}]
```

The option **AspectRatio→Automatic** is to make sure that the scales on the vertical and horizontal axes are the same. The option **AxesOrigin →{0,0}** is to make sure that the vertical and horizontal axes meet each other at the origin $(0, 0)$.

To get better visualization, we will plot in motion. That is, as we trace points on the line $z = 1 + it$ from $t = 0$ to $t = 1$, we want to see how the images are drawn out. This can be done as follows. For each $s > 0$, we sketch the image of line $z = 1 + it$ for $t \in [0, s]$. Then allows s to increase from a small positive number, say 0.1, to 1 to see how the images are drawn out (Figure 3). We do so by using the command **Manipulate**.

```
p[s_] :=
 ParametricPlot[ReIm[1 + t*I], {t, 0, s}, AxesOrigin -> {0, 0},
  PlotRange -> {{0, 2}, {-1, 2}}]
q[s_] :=
 ParametricPlot[ReIm[f[1 + t*I]], {t, 0, s}, AxesOrigin -> {0, 0},
  PlotRange -> {{0, 3}, {0, 3}}]
Manipulate[{p[s], q[s]}, {s, .1, 1}]
```



Figure 2



Figure 3

The option **PlotRange** $\rightarrow$ **{{a,b},{c,d}}** is to make sure that we view the graph only within the region $a \leq x \leq b$, $c \leq y \leq d$. Without specifying PlotRange, the viewing region may change as s varies, which can cause annoyance.

In Figure 3, the plot on the left shows the line $x = 1$ as we draw it upward; the plot on the right shows the image of the line under the exponential function.

Can you adjust the above code to draw the image of the line $z = 1 + it$ as t moves from 0 to 10? What does the shape look like?